

7. STRUKTURE PODATAKA

U mnogim primjenama javlja se veći broj elementarnih podataka koji čine izvjesnu logičku cjelinu. Za ovakve kolekcije podataka kažemo da obrazuju **strukturu podataka**.

Jednu strukturu podataka smo već upoznali u LBu, a to je slog (ulazni i izlazni slog), koji služi za komunikaciju računara s ulaznim i izlaznim uređajima.

Drugu strukturu podataka u LBu predstavlja **niz**. Niz je kolekcija podataka istog tipa. To je struktura podataka koja odgovara strukturama podataka u matematici, kao što su vektori i matrice.

7.1 Ime niza i elementi niza

Jednoznačna identifikacija niza vrši se pomoću imena niza. Ime niza se definiše na isti način kao i imena drugih objekata u LB jeziku (npr. kod promjenjivih). Dakle, to može biti izabrana riječ, pri čemu prvi znak mora biti slovo, a ostali znaci mogu biti slova ili cifre.

Svi podaci koji obrazuju niz moraju biti istog tipa i dužine. Saglasno tipu podataka u nizu mora se deklarisati i ime niza.

Slijedeće važno pitanje za rad s nizovima jeste način referisanja na elementarne podatke koji čine niz. Ovo se ostvaruje navođenjem indeksâ niza pomoću kojih se jednoznačno može ukazati na svaki element niza. Indeksi niza se navode između zagrada, iza imena niza i međusobno se razdvajaju zarezima. Broj indeksa koji je potreban za ukazivanje na određeni element niza određuje dimenzionalnost niza. Po ovoj osobini, nizovi mogu biti jednodimenzionalni ili višedimenzionalni. Primjeri:

- ALFA(4) – definiše četvrti element jednodimenzionalnog niza ALFA,
- B(2,3) – definiše element (2,3) dvodimenzionalnog niza B.

Ime niza, sa indeksima koji ukazuju na određeni element niza, zovemo **indeksna promjenljiva** (ALFA(4), B(2,3)).

Indeksne promjenljive se mogu javiti u izrazima kao argumenti izraza, ili to mogu biti promjenljive na lijevoj strani znaka jednakosti, kojima se dodjeljuju izračunate vrijednosti, kao na primjer:

$$A(4,6) = (3.5 + B(1)) / C(8,10)$$

7.2 Dimenzije niza

Za registrovanje niza u memoriji računara neophodno je rezervisati potreban memorijski prostor. U tu svrhu mora se navesti broj elemenata svakog niza koji se koristi u programu u posebnoj naredbi koja se piše u obliku:

Dim ime(dimenzija), ime(dimenzija), ...

gdje su:

- ime – ime niza,
- dimenzija – cijeli brojevi koji određuju maksimalne vrijednosti indeksâ, međusobno razdvojeni zarezima.

Granice se zadaju u obliku cijelih brojeva. rijednost 1. Tako naredba:

Dim A(20), B(100), C(3,5)

deklariše nizove A i B kao jednodimenzionalne, a niz C kao dvodimenzionalni. Niz A će imati 20 elemenata: A(1), A(2), ... , A(20). Niz C sadrži elemente:

$$C(3,5) = \begin{bmatrix} C(1,1) & C(1,2) & C(1,3) & C(1,4) & C(1,5) \\ C(2,1) & C(2,2) & C(2,3) & C(2,4) & C(2,5) \\ C(3,1) & C(3,2) & C(3,3) & C(3,4) & C(3,5) \end{bmatrix}$$

Svaki niz, deklarisan u programu, zauzima kompaktan memorijski prostor, a to znači da se za registrovanje elemenata niza koriste susjedne memorijske lokacije. Za registrovanje gore deklarisanog niza A(20) koristi se 20 memorijskih lokacija, pri čemu se elementi niza redaju redom A(1), A(2), ... , A(20).

Primjenom višestrukih petlji, višedimenzionalni nizovi se registruju tako što se indeksi vanjskih petlji mijenjaju brže od unutarnjih, tako da se na primjer elementi dvodimenzionalnih nizova registruju red po red. Za svaku vrijednost indeksa vanjske petlje, izredaju se sve vrijednosti indeksa unutarnje petlje, prije nego se pređe na novu vrijednost indeksa vanjske petlje. Unose li se elementi niza putem tastature i ekrana, brojevi koji se budu redom unosili (na način kako je to ranije objašnjeno kod rada sa brojevima veličinama), pohranjivaće se u memorijske lokacije elemenata prema redoslijedu naznačenom uz elemente niza C.

Indeksi niza mogu biti aritmetički izrazi čije vrijednosti ne smiju biti manje od donje, niti veće od gornje granice odgovarajuće dimenzije deklarisane u naredbi DIM. Tako na primjer u naredbi:

$$A(I) = 4*B(3+J)$$

promjenljiva I mora imati vrijednost iz intervala [1,20], a izraz 3+J vrijednosti iz intervala [1,100].

7.3 Nizovi u ulazno/izlaznim (U/I) listama

U programima je često potrebno da se elementima niza dodijele vrijednosti sa ulaza, ili da se izdaju vrijednosti elemenata niza. Ovo zahtijeva da se nizovi javljaju u listovima naredbi ulaza, odnosno izlaza. Elementi niza se mogu pojaviti u U/I listama na dva načina:

- Elementi niza ne slijede jedan za drugim u redoslijedu u kojem se čuvaju u memoriji. U ovom slučaju elementi niza se navode u U/I listi kao indeksne promjenljive:

Input A(4), A(2), A(8), A(15)

što znači da će navedenim elementima biti dodijeljene vrijednosti sa ulaza.

- Elementi niza slijede jedan za drugim ili slijede po određenom zakonu. Navođenje ovakvih elemenata niza u U/I listama može se pisati bez nabranja svih elemenata ponaosob. Za pojedine slučajeve jednodimenzionalnih, dvodimenzionalnih i višedimenzionalnih nizova, detaljno se o ovom slučaju govori u nastavku teksta.

Jednodimenzionalni nizovi

Za navođenje elemenata jednodimenzionalnog niza u U/I listi, pri čemu elementi niza slijede određenu zakonitost, može se koristiti zapis:

```
For I = 1 To M
    Input ime(I)
Next I
```

gdje su:

| | | |
|-----|---|--------------------------------------|
| ime | - | ime niza, |
| I | - | cjelobrojna promjenljiva, |
| M | - | maksimalne vrijednosti indeksa niza. |

Gornji zapis ima isti efekat kao da su navedeni elementi niza u redoslijedu:

```
ime(1), ime(2), ime(3), ... , ime(M)
```

U sljedećem primjeru P701 dat je program kojim se određuje najveći element jednodimenzionalnog niza X.

```
'Program P701

Dim X(50)

Input "Broj elemenata niza:"; N
Print "Unesite elemente niza:"
For I = 1 To N
    Input X(I)
Next I

MAX = X(1)
For I = 2 To N
    If (MAX < X(I)) Then MAX = X(I)
Next I

Print "NAJVECI ELEMENT JE:"; MAX

End
```

U nastavku su data još dva primjera rada sa jednodimenzionalnim nizovima. Prvi, u kojem je zadatak sortirati u rastućem redoslijedu zadani jednodimenzionalni niz, i drugi, kojim se izračunava vrijednost polinoma, čiji su stepen i koeficijenti zadani na početku programa. Kod drugog zadatka koristilo se izrazom:

$$P(X) = A(N) \cdot X^n + A(N-1) \cdot X^{n-1} + \dots + A(1) \cdot X + A(0)$$

```

' Program SORT

DIM X(20), Y(20)

' UCITAVANJE NIZA ZA SORTIRANJE U RASTUCEM REDOSLIJEDU
10 Input "BROJ ELEMENATA NIZA (<=20): "; N
    IF (N < 1) Then GOTO 10
    IF (N > 20) Then GOTO 10

Print "UNESI ELEMENTE NIZA:"
For I = 1 To N
    Input X(I)
Next I

' FORMIRANJE NIZA Y ISTOVJETNOG NIZU X
For I = 1 To N
    Y(I)=X(I)
Next I

' SORTIRANJE NIZA
For I = 1 To N-1
    For J = I+1 To N
        IF (Y(I) > Y(J)) THEN
            POM=Y(I)
            Y(I)=Y(J)
            Y(J)=POM
        END IF
    Next J
Next I

' STAMPANJE POLAZNOG I SORTIRANOG NIZA
Print
Print "POLAZNI NIZ      SORTIRANI NIZ"

For I = 1 To N
    Print X(I),Y(I)
Next I

End

```

```

' Program POLI

DIM A(20)

Print "IZRACUNAVANJE VRIJEDNOSTI POLINOMA DATOG U OBLIKU:"
Print "P(X) = A(N)X^N+A(N-1)X^(N-1)+...+A(1)X + A(0)"
Print
Input "UNESI STEPEN POLONOMA: "; N

```

```

Print "UNESI KOEFICIJENTE POLINOMA A(N) . . . A(0) : "
For I = N To 0 Step -1
    Input A(I)
Next I

Input "UNESI VRIJEDNOST ARGUMENTA X: " ; X

' IZRACUNAVANJE VRIJEDNOSTI POLINOMA
P=0
For I = N To 0 Step -1
    P = P + A(I) * X ^ I
Next I

' STAMPANJE VRIJEDNOSTI POLINOMA
Print "VRIJEDNOST POLINOMA JE P = " ; P

End

```

Dvodimenzionalni nizovi

Elementi dvodimenzionalnih nizova mogu se takođe pojaviti u listama ulaznih i izlaznih naredbi. Kada se javljaju nesusjedni elementi, tada se mogu jednostavno navesti ponaosob:

```
Print A(35,10), B(18,4), C(3,3)
```

Kada elementi dvodimenzionalnog niza u U/I listi slijede jedan za drugim, ili slijede po određenom zakonu, tada se u listi može navesti zapis:

```

For I = 1 To M
    For J = 1 To N
        Input/Print ime(I,J)
    Next J
Next I

```

koji proizvodi slijedeći redoslijed elemenata:

```
ime(1,1), ime(1,2), ime(1,3), ... , ime(1,N), ime(2,1), ime(2,2), ... , ime(2,N), ... ,
ime(M,1), ime(M,2), ime(M,3), ..., ime(M,N)
```

Tako zapis:

```

For I = 1 To 5
    For J = 1 To 10
        Print A(I,J)
    Next J
Next I

```

proizvodi slijedeći redoslijed elemenata niza A u listi:

```
A(1,1), A(1,2), A(1,3), A(1,4), A(1,5), ... , A(1,10), A(2,1), A(2,2), A(2,3), A(2,4),
A(2,5), ... , A(2,10), A(3,1), A(3,2), A(3,3), A(3,4), A(3,5), ... , A(3,10), ... , A(5,1),
A(5,2), A(5,3), A(5,4), A(5,5), ... , A(5,10)
```

odnosno:

$$A(5,10) = \begin{bmatrix} A(1,1) & A(1,2) & \dots & A(1,5) & \dots & A(1,10) \\ A(2,1) & A(2,2) & & A(2,5) & & A(2,10) \\ A(3,1) & A(3,2) & & A(3,5) & & A(3,10) \\ A(4,1) & A(4,2) & & A(4,5) & & A(4,10) \\ A(5,1) & A(5,2) & \dots & A(5,5) & \dots & A(5,10) \end{bmatrix}$$

Primjer P702

Program za sabiranje dviju matrica. Na ulazu se zadaju matrice A(3,6) i B(3,6), a u programu se izračunava:

$$C(3,6) = A(3,6) + B(3,6).$$

Matrice su u programu definisane kao dvodimenzionalni nizovi, a njihovo unošenje vrši se red po red. Sabiranje elemenata matrica A i B vrši se takođe red po red, kao i izdavanje elemenata rezultujuće matrice C.

```
'Program P702
DIM A(3,6), B(3,6), C(3,6)

Print "UNESITE RED PO RED MATRICU A:"
For I = 1 To 3
    For J = 1 To 6
        Input A(I,J)
    Next J
Next I

Print "UNESITE RED PO RED MATRICU B:"
For I = 1 To 3
    For J = 1 To 6
        Input B(I,J)
    Next J
Next I

For I=1 To 3
    For J=1 To 6
        C(I,J) = A(I,J) + B(I,J)
    Next J
Next I

Print
Print "ZBIR MATRICA A i B JE:"
I = 0
20 I = I + 1
    Print
    For J = 1 To 6
        Print C(I,J); " ";
    Next J
```

```
If (I < 3) Then Goto 20
Print
End
```

U LBu se elementi dvodimenzionalnog reda – matrice mogu unositi samo jedan po jedan, uz potvrđivanje unosa naredbom <Enter>. Štampanje rezultujuće matrice C već se, primjenom naredbe Print, može realizovati štampanjem iste red po red, na način kako je to pokazano u prethodnom primjeru.

Primjer P703 ilustruje rad sa nizovima čiji su elementi znakovni podaci. Dvodimenzionalnim nizovima A i B se sa ulaza dodjeljuju znakovni podaci i određuju elementi niza C, tako da se iz elemenata matrice A uzimaju prva dva znaka, kojima se dopisuju posljednja dva znaka odgovarajućeg elementa niza B.

```
'Program P703
DIM A$(3,3), B$(3,3),C$(3,3)

Print "Unesite matrice A i B:"
For I = 1 To 3
    For J = 1 To 3
        Input A$(I,J), B$(I,J)
    Next J
Next I

For I = 1 To 3
    For J = 1 To 3
        C$(I,J) = Mid$(A$(I,J),1,2) + Mid$(B$(I,J),3,4)
    Next J
Next I

Print
Print "Matrica C ima slijedeće elemente:"
I = 0
20 I = I + 1
    Print
    For J = 1 To 3
        Print C$(I,J); " ";
    Next J
    If (I < 3) Then Goto 20

Print
End
```

Napomena: elementi nizova A i B se unoše u parovima, tako što se prvo unosi odgovarajući element niza A (na primjer A(1,1), a zatim element niza B sa istim indeksima (dakle B(1,1)).

Primjer unosa ulaznih podataka:

A(1,1) ABCD

B(1,1) EFGH

A(1,2) IJKL

B(1,2) MNOP

A(1,3) QRST

B(1,3) UVWX

A(2,1) 1234

B(2,1) 5678

A(2,2) 9012

B(2,2) 3456

A(2,3) 7890

B(2,3) 1234

A(3,1) **++

B(3,1) **++

A(3,2) **++

B(3,2) **++

A(3,3) **++

B(3,3) **++

Rezultat će imati slijedeći izgled:

Matrica C ima slijedeće elemente:

ABGH□IJKL□MNOP□QRST

1278□9056□7834

* * ++□* * ++□* * ++